Лабораторна робота №2

Знайомство із середовищем розробки програм Microsoft Visual Studio

Мета: Отримати перші навички роботи з Microsoft Visual Studio для створення програм, написаних мовою асемблера, а також вивчити команди MOV та CPUID.

Завдання:

- 1. Створити у середовищі MS Visual Studio проект з ім'ям **Lab2**. Встановити необхідні параметри проекту опції середовища розробки програм.
- 2. Написати вихідний текст програми на асемблері, додати файл вихідного тексту у проект. Зміст вихідного тексту згідно з варіантом завдання.
- 3. Скомпілювати вихідний текст і отримати виконуємий файл програми.
- 4. Перевірити роботу програми. Налагодити програму.
- 5. Отримати дизасембльований текст машинного коду і проаналізувати його.

Теоретичні відомості

Асемблер i Microsoft Visual Studio

Система Microsoft Visual Studio призначена для розробки програм на різноманітних мовах програмування – С#, Visual Basic, С++, а також асемблері.

Необхідно відзначити, що оскільки у середовищі MS Visual Studio для компіляції асемблерних файлів використовується компілятор MASM (подібний до MASM32), то синтаксис вихідних текстів подібний до розглянутого вище у лабораторній роботі №1. Таким чином, теоретичні положення щодо мови асемблеру, викладені у розділі теоретичних відомостей для попередньої лабораторної роботи №1, є чинними і для роботи №2.

Ця робота присвячена розробці програм типу Win32 — 32-бітних програм, які працюють у відповідному середовищі операційної системи Windows. Скелет вихідного тексту на асемблері для таких програм повністю повторює вже розглянутий вище у попередній роботі. Більше того, як передбачається, можуть бути використані деякі файли зі складу пакету MASM32 — а саме бібліотек функцій API Windows.

Команда MOV

Ця команда часто використовується у програмах на асемблері. Вона виконує копіювання даних. Команда MOV має два операнди:

```
то Куди, Джерело
```

Операнд **джерело** повинен вказувати, звідки взяти інформацію. Перший операнд вказує, куди записати інформацію. Наприклад:

```
mov ecx, eax
```

означає скопіювати дані з регістру ЕАХ у регістр ЕСХ. У наступному рядку

```
mov ax, 5
```

запрограмований запис числа 5 у регістр АХ. У якості другого операнду записане безпосередньо числове значення. Такі значення зберігаються у пам'яті, тому фактично виконується копіювання типу "пам'ять → регістр".

Певна кількість байтів джерела повинна записуватися у відповідне за розміром місце. Приклади помилок:

```
mov al, edx;помилка: з 32-бітового у 8-бітовий регістрmov ax, ecx;помилка: з 32-бітового у 16-бітовий регістрmov eax, cx;помилка: з 16-бітового у 32-бітовий регістр
```

Наступні приклади:

```
mov al, 5 ; AL = 00000101 (8 біт)
mov ax, 5 ; AX = 00000000000101 (16 біт)
mov eax, 5 ; EAX = 000000000000000000000000101 (32 біти)
```

Тут помилок немає – у регістри записується відповідна кількість бітів значення, яке може представлятися як 8-бітовим, так і 16-, або 32-бітовим двійковим кодом. У той же час запис

```
mov al, 259 ; помилка: у 8-бітовий регістр число 259 не можна
```

неприпустимий – для представлення числа 259 восьми бітів замало. Асемблер при компіляції видаєть помилку.

Можна сказати, що форма запису на асемблері

```
mov a, b
```

означає операцію присвоювання у мові програмування високого рівня:

```
a = b
```

Можна розглянути присвоєння значення однієї перемінної іншій, запрограмоване мовою C/C++:

```
long a, b;
a = b;
```

Записати відповідний код на асемблері можна спробувати так:

```
.data
a dd ? ; створення неініціалізованої перемінної а
b dd ? ; створення неініціалізованої перемінної b
.code
mov a, b ; помилка, так не можна
```

Перемінні а та $b \in oб'єктами у пам'яті. Одною командою MOV копіювати з пам'яті у пам'ять не можна. Можна, наприклад, так:$

```
mov eax, b ; pericтр EAX у якості посередника mov a, eax
```

Необхідно зазначити, що у наведених вище рядках використання імен перемінних у якості операндів команд MOV дещо спрощує синтаксис, проте ховає те, що насправді замість імен використовуються адреси відповідних перемінних. Більш адекватний програмний код того, що насправді виконується, можна отримати у вікні дизасемблера.

```
mov eax, dword ptr [b]
mov dword ptr [a], eax
```

У мові асемблера квадратні дужки означають, що всередині них міститься вказівник – той, хто зберігає адресу пам'яті.

Синтаксис мови асемблера MASM достатньо гнучкий. Запис

mov eax, b

можна вважати дещо спрощеним варіантом, аніж

mov eax, [b]

який, у свою чергу, є спрощенням від:

mov eax, dword ptr [b]

Остання форма запису ϵ найбільш коректною.

Розглянемо наступний приклад. Створимо масив з чотирьох 32-бітових елементів – елементів типу DWORD:

```
.data
M dd 4 dup(?)
```

Елементи масиву розташовуються у пам'яті послідовно. Загалом для масиву потрібно 16 байтів пам'яті (рис. 1)




Рис. 1. Масив з чотирьох елементів типу DWORD

Запишемо у другий елемент якесь значення, наприклад:

mov dword ptr[M+4], 89ABCDEFh

Результат на рис. 2.





Рис. 2

Ще приклад:

mov dword ptr[M+7], 10CCABBAh

Результат на рис. 3.



Приклад запису двохбайтового значення у пам'ять

mov word ptr[M+8], 4352h

Результат на рис. 4.




Рис. 4. Запис двох байтів у пам'ять

Приклад запису однобайтового значення у пам'ять

mov byte ptr[M+9], 2Ah

Результат на рис. 5.




Рис. 5. Запис одного байту у пам'ять

Команда CPUID

Ця команда призначена для ідентифікації процесора та отримання відомостей про його властивості. Для того, щоб вказати, які саме відомості потрібні, треба записати у регістр EAX деяке значення — параметр для команди. Результати роботи команди CPUID процесор записує у регістри EAX, EBX, ECX EDX.

Для того, щоб правильно користуватися командою CPUID, потрібно дотримуватися певної послідовності роботи програми. Можна спочатку перевірити — а чи можливо взагалі користуватися цією командою? Для цього треба перевірити значення 21-го біту регістру EFLAGS. Проте, у цій лабораторній роботі можна вважати це зайвим.

Отримання відомостей за допомогою команди CPUID робиться у декілька кроків. Спочатку треба виконати цю команду із значенням параметру 0:

mov eax,0
cpuid

Після виконання такого коду, у регістрі ЕАХ міститься значення, яке означає максимально можливе значення параметру, яке можна використати для отримання базових відомостей. Якщо намагатися викликати команду CPUID із параметром більше зазначеного вище максимального, то результатом будуть усі нулі (це не стосується параметрів так званих розширених відомостей).

Після виконання команди CPUID із параметром 0 процесор також записує у регістри EBX, ECX та EDX коди символів імені процесора. Для процесорів Intel це буде "GenuineIntel", для процесорів AMD — "AuthenticAMD" тощо. Необхідно відзначити, що 12 символів записуються четвірками у такому порядку — спочатку EBX, потім EDX, останні чотири у регістрі ECX.

Наступним кроком у програмі буде виконання команди з параметром 1:

mov eax,1 cpuid

Після виконання цієї команди у регістри EAX, EBX, ECX та EDX буде записано інформацію про сімейство процесорів, модель та деякі інші відомості.

Наступним кроком у програмі буде виконання команди з параметром 2:

mov eax,2 cpuid

і так далі, наскільки можливо. Як вже вказувалося вище, максимально можливе значення параметру стає відомим після виконання команди з параметром 0. Це для базового набору відомостей. Проте, є ще так звані, розширені відомості, які відповідають функціям на основі команди CPUID з параметром 80000000h і більше. Для того, щоб дізнатися максимальне значення для параметру розширених функцій, потрібно виконати:

mov eax,80000000h cpuid

У результаті виконання цього у регістрі ЕАХ буде деяке значення, наприклад, 80000008h. Можна послідовно виконувати команди CPUID із значеннями параметрів від 80000001h до 80000008h для отримання відповідних відомостей. Наприклад, після виконання коду:

```
mov eax,80000008h cpuid
```

у регістрі ЕАХ буде інформація про максимальну можливу розрядність адрес пам'яті даного процесора.

Вичерпна інформація щодо CPUID міститься у документі "Intel® 64 and IA-32 Architectures Software Developer's Manual Volume 2A: Instruction Set Reference", який можна завантажити через інтернет по адресі http://www.intel.com.

У підсумку, виконання ланцюжка команд CPUID можна відобразити так:

```
;---базові функції---
mov eax, 0; noyatok
cpuid
            ;зберігання значень EAX,EBX,ECX,EDX
mov eax,1
cpuid
            ;зберігання значень EAX,EBX,ECX,EDX
             ;якщо можливо, то інші функції для ЕАХ>1
;---розширені функції---
mov eax,80000000h
cpuid
             ;зберігання значень EAX, EBX, ECX, EDX
mov eax,8000001h ;якщо можливо
cpuid
             ;зберігання значень EAX, EBX, ECX, EDX
             ;якщо можливо, то інші функції для EAX>8000001h
```

Для зберігання значень регістрів EAX, EBX, ECX та EDX можна їх записувати у масив-вектор, наприклад:

```
.data
res dd 256 dup(0)

.code
mov eax, 0
cpuid
mov dword ptr[res], eax
mov dword ptr[res+4], ebx
mov dword ptr[res+8], ecx
mov dword ptr[res+12], edx
. . . ;у подібний спосіб і для інших СРИІО
```

Проте, зберігання четвірок 32-бітових значень можна запрограмувати, як здається, і по-іншому.

Вивід інформації

Текстова інформація може відображатися достатньо простими засобами – викликом стандартних діалогових вікон MessageBox. Для цього достатньо сформувати у двох буферах потрібний текст: основний і текст заголовку:

```
.data
    Text db 256 dup(0)
    Caption db 32 dup(0)
.code
    · · · ;формування вмісту буферів Техт, Caption
    invoke MessageBoxA, 0, ADDR Text, ADDR Caption, 0
```

Як сформувати вміст буферів тексту? Для даної роботи потрібно виводити значення, отримані командами CPUID. Числові значення можна показувати у шістнадцятковому коді. Для цього потрібно перетворювати 32-бітові значення типу DWORD у 8 символів шістнадцяткових цифр. Таке перетворення можна виконати за допомогою окремої процедури **DwordToStrHex**. Процедура викликається наступним чином:

```
push ... ;числове 32-бітове значення
push offset ... ;адреса буферу тексту
call DwordToStrHex
```

Вихідний текст процедури **DwordToStrHex** пропонується нижче:

```
;ця процедура записує 8 символів НЕХ коду числа
;перший параметр - 32-бітове число
;другий параметр - адреса буфера тексту
DwordToStrHex proc
  push ebp
  mov ebp, esp
  mov ebx,[ebp+8] ; другий параметр
  mov edx,[ebp+12] ; перший параметр
  xor eax, eax
  mov edi,7
@next:
  mov al,dl
  and al,0Fh
                     ;виділяємо одну шістнадцяткову цифру
;так можна тільки для цифр 0-9
   add ax,48
   cmp ax,58
   jl @store
   add ax,7
                     ;для цифр A,B,C,D,E,F
@store:
  mov [ebx+edi],al
   shr edx,4
   dec edi
   cmp edi,0
   jge @next
   pop ebp
   ret 8
DwordToStrHex endp
```

Формування символів шістнадцяткових кодів для значень EAX, EBX, ECX, EDX, збережених у масиві **res**, може бути запрограмоване, наприклад, так:

```
.data
 res dd 256 dup(0)
 Text db 'EAX=xxxxxxxx',13,10,
         'EBX=xxxxxxx',13,10,
         'ECX=xxxxxxxx',13,10,
         'EDX=xxxxxxxx',0
 Caption db "Результат CPUID 0",0
.code
                         ; інший програмний код (заповнення масиву res)
 . . .
 push [res]
                         ;значення регістру EAX з масиву res
 push offset [Text+4] ;адреса, куди записуються 8 символів
 call DwordToStrHex
 push [res+4]
                        ;значення регістру EBX з масиву res
 push offset [Text+18]
 call DwordToStrHex
                        ;значення регістру ECX з масиву res
 push [res+8]
 push offset [Text+32]
 call DwordToStrHex
 push [res+12]
                        ;значення регістру EDX з масиву res
 push offset [Text+46]
 call DwordToStrHex
 invoke MessageBoxA, 0, ADDR Text, ADDR Caption, 0
```

Таким чином, можна отримати, наприклад, такий текст (рис. 1):



Рис. 1. Відображення значень після виконання команди CPUID

Як окремий випадок можна відзначити вивід імені процесора, яке записується у регістри EBX, ECX та EDX після виконання команди з параметром CPUID 0. Особливість полягає у тому, що у вказані регістри тут записуються безпосередньо однобайтові ASCII коди символів — по чотири символи в регістр. Тоді ці коди символів можна прямо вставити у відповідні байти текстового буферу. Наприклад:

```
.data
Vendor db 16 dup(0)
CaptionVendor db "CPUID 0 Vendor string",0
... ;інші дані

.code
mov eax, 0
cpuid
mov dword ptr[Vendor], ebx
mov dword ptr[Vendor+4], edx
mov dword ptr[Vendor+8], ecx
... ;інший програмний код

invoke MessageBoxA, 0, ADDR Vendor, ADDR CaptionVendor, 0
```

У діалоговому вікні буде відображатися наступне (рис. 2):



Рис. 2. Один з результатів виконання команди CPUID

Стосовно конфіденційності. Команда СРUID дозволяє встановити приналежність процесора тільки до певного типу. Індивідуально процесори не ідентифікуються, хоча колись така можливість розглядалася, проте потім, як здається, розробники процесорів архітектури х86 від цього відмовилися.

Порядок виконання роботи та методичні рекомендації

При вивченні системи MS Visual Studio треба звернути увагу на те, що ϵ різні версії цієї системи відповідно різним рокам випуску та мовам інтерфейсу. Україномовної версії поки що нема ϵ . Опис порядку виконання роботи та методичні рекомендації тут конкретизовані для російськомовного варіанту MS Visual Studio 2010.

1. Створення проекту у середовищі MS Visual Studio

При створенні проекту чисто на асемблері потрібно враховувати особливості середовища розробки програм. Першою особливістю ϵ те, що для проектів MS Visual Studio у переліку мов програмування асемблер відсутній. Рекомендація: вкажіть у списку мов проект Visual C++.

Виберіть меню "Файл – Создать – Проект", і далі у діалоговому вікні "Создать проект" виберіть Visual C++, а потім – Проект Win32 (рис.3):




Рис. 3.

Потрібно ввести ім'я програми. Нехай ім'я програми та ім'я рішення буде, наприклад, Lab2. Вводимо його і натискуємо ОК.

З'являється вікно "Мастер приложений Win32". Нам потрібно визначити деякі параметри проекту. Натискуємо кнопку "Далее >". З'являється наступне діалогове вікно (рис. 4):




Рис. 4. Основні параметри проекту програми Lab2

Вибираємо **порожній проєкт**, поставивши відповідну позначку у групі "Дополнительные параметры". Натискуємо кнопку "Готово". Середовище Visual Studio переходить у стан роботи з новоствореним проєктом. У вікні "Обозреватель решений" показується такий список (рис. 5):




Рис. 5.

Оскільки ми створили порожній проект, то жодних вихідних текстів у вказаних розділах немає. Потрібно додавати тексти програми у папку **Файли исходного кода** та у інші папки (якщо потрібно). Проте, спочатку необхідно дещо налагодити робоче середовище, інакше асемблерні тексти Visual Studio сприймати не буде.

2. Налагодження підтримки мови асемблера

У вікні "Обозреватель решений" встановіть курсор на імені проекту (Lab2) і клацніть правою кнопкою миші. У спливаючому меню виберіть пункт "Настройки построения..." (рис. 6).




Рис. 6.

Далі з'явиться діалогове вікно (рис. 7)




Рис. 7.

Виберіть пункт "masm", та зробіть позначку у квадратику. Натисніть ОК. Таким чином у проекті активується вбудований у Visual Studio асемблер MASM, який буде компілювати файли вихідних текстів *.asm.

3. Додавання вихідних текстів у проект

Нам потрібно додати у проєкт файл вихідного тексту на асемблері. Будемо вважати, що тексту поки що немає і ми плануємо його писати безпосередньо у середовищі Visual Studio. Вкажіть курсором розділ "Файлы исходного кода", клацніть правою кнопкою миші, і потім - "Создать элемент" (рис. 8):




Рис. 8.

З'являється наступне вікно (рис. 9):




Рис. 9.

У цьому вікні в списку типів файлів немає файлів асемблера. Потрібно вибрати "Текстовый файл (.txt)". Вводимо ім'я файлу – Lab2.asm і натискуємо кнопку "Добавить".

У середовищі Visual Studio відкривається нове порожнє вікно з ім'ям lab2.asm, у якому ми будемо писати текст програми. Наприклад, такий:

```
.586
.model flat, stdcall

include \masm32\include\kernel32.inc
include \masm32\lib\kernel32.lib
includelib \masm32\lib\kernel32.lib
includelib \masm32\lib\user32.lib

.data
    Caption db "Я програма на асемблері",0
    Text db "Здоровеньки були!",0

.code
main:
    invoke MessageBoxA, 0, ADDR Text, ADDR Caption, 0
invoke ExitProcess, 0
end main
```

Ця програма виводить текст-вітання "Здоровеньки були!" у діалоговому вікні із заголовком "Я програма на асемблері".

4. Підключення бібліотек

Програма викликає дві функції Windows, які доступні через інтерфейс API Win32:

- **MessageBoxA** для показу простого діалогового вікна. Ця функція міститься у системній DLL **user32.dll**. Для побудови проекту лінкер використовує бібліотеку імпорту **user32.lib**.
- **ExitProcess** цю функцію повинна викликати будь-яка програма для коректного завершення. Вона викликається з системної DLL **kernel32.dll**. Для побудови проекту використовується бібліотека імпорту **kernel32.lib**.

Підключення вказаних бібліотек до проекту описується рядками

```
includelib \masm32\lib\kernel32.lib
includelib \masm32\lib\user32.lib
```


Для компіляції викликів функцій **MessageBoxA та ExitProcess** потрібно знати, які вони мають аргументи. Опис аргументів міститься у заголовочних файлах *.inc, які включені у текст директивами include:

```
include \masm32\include\kernel32.inc
include \masm32\include\user32.inc
```

Де взяти файли kernel32.inc, kernel32.lib, user32.inc та user32.lib? Серед файлів Visual Studio їх немає. Вказані файли можна взяти з комплекту MASM32 (дивіться лабораторну роботу №1). Потрібно створити на диску, на якому записується поточний проект, наприклад, диску D, папки d:\masm32\lib та d:\masm32\include і записати туди вказані вище файли.

5. Підготовка до компіляції та лінкування

Потрібно налагодити лінкер. У вікні властивостей проекту виберіть розділ "Свойства конфигурации — Компоновщик — Дополнительно". У рядку "Точка входу" запишить ім'я процедури, яка буде викликатися від початку роботи нашої програми (рис. 9). У програмі може бути єдиною процедура, яка завершується END з міткою-ім'ям процедури. У тексті Lab2.asm така процедура позначена **main** і вона буде точкою входу:






Рис. 10

Примітка. Може так траплятися, що лінкер буде сам знаходити потрібну процедуру, навіть якщо у властивостях проекту не вказувати ім'я процедури –

точку входу. Проте, якщо лінкер видасть помилку, то однією з причин може бути невизначеність точки входу для поточного проекту.

6. Компіляція та створення виконуємого файлу програми

Для створення виконуємого файлу програми потрібно спочатку скомпілювати вихідний текст, а потім лінкер запише файл — lab2.exe. Потім викликати програму на виконання. Усе це можна зробити одразу через меню "Отладка — Начать отладку".

У разі успішної компіляції, лінкування, запису файлу lab2.exe і виклику програми, ця програма запрацює і покаже вікно-вітання (рис. 11):




Рис. 11. Вигляд вікна програми lab2 на тлі вікон MS Visual Studio

Натисніть кнопку "ОК" – програма закінчить свою роботу. Про цю подію середовище MS Visual Studio у вікні "Вывод" нас повідомить так:

Программа "[0x894] Lab2.exe: Машинный код" завершилась с кодом 0 (0x0).

7. Debug- та Release-версії програми

За умовчанням при створенні нового проєкту середовище Visual Studio відпрацьовує Debug-версію нашої програми. Що це означає? У машинний код вставляються команди, які дозволяють контролювати роботу програми, налагоджувати її. У робочій папці створюється вкладена папка Debug – у нашому проєкті це E:\Stud\Lab2\Debug у яку після кожного успішного сеансу

лінкування записується файл lab2.exe. Debug-версія програми є деяким тимчасовим зразком, який має сенс тільки на етапі розробки та налагодження. Якщо програміст хоче розповсюджувати власну програму, то Debug-версія для цього не дуже придатна з багатьох причин.

Фінальна версія ехе-файлу програми, яка призначена для розповсюдження і використання на інших комп'ютерах, зветься Release-версією. Щоб налаштувати середовище на створення цієї версії, виберіть меню "Построение - Конфігурація". Далі у наступному вікні виберіть "Release" (рис. 12).




Рис. 12. Зміна конфігурації Debug на Release

Середовище починає працювати трохи по-іншому. Набір властивостей компілятора, лінкера тощо тут зветься конфігурацією. Виконуємий файл буде записуватися у вкладену папку Release - у нашому проекті це буде E:\Stud\Lab2\Release. Порівняйте файли lab2.exe у папці Release та у папці Debug.

Необхідно враховувати відмінність багатьох параметрів роботи середовища для різних конфігурацій. Це стосується також і параметрів (опцій) роботи компілятора та лінкера.

8. Налагодження програми, огляд та аналіз машинного коду

Зазвичай програму потрібно налагоджувати, виправляти помилки, вдосконалювати. У цьому може допомогти налагоджувач (debugger).

Налаштуйте середовище Visual Studio на Debug-конфігурацію. Потім потрібно встановити точки зупинок (*точки останова рос., breakpoints англ.*). Точки зупинки ставляться у лівому стовпчику вікна вихідного тексту. Нехай нам потрібно відстежити роботу програми від початку - ставимо точку на першому рядку (рис. 13):

```
    .code
        main:
            invoke MessageBoxA, 0, ADDR Text, ADDR Caption, 0
            invoke ExitProcess, 0
            end main
```

Рис. 13. Точка зупинки

Розпочинаємо налагодження, натиснувши клавішу F5. Програма не зупиняється, немов би не помічає точку зупинки. У чому справа? Порада: якщо точка зупинки на першій команді, то розпочинайте налагоджування, натиснувши клавішу F10.

Знову розпочнемо налагодження. Натиснемо клавішу F10. Тепер процес зупиняється у потрібній точці. Для перегляду значень регістрів відкрийте відповідне вікно через меню "Отладка — Окна — Регистры". Для аналізу створеного компілятором машинного коду відкрийте вікно дизасемблерного коду через меню "Отладка — Окна — Дизассемблированный код".

У вікні "Дизассемблированный код" відображається машинний код, який згенерував компілятор. Це той код, який насправді виконує програма. Проте у вікні дизасемблера надається не власне машинний код (це множина двійкових чисел — кодів команд та даних), а текст у вигляді рядків на мові асемблеру як результат зворотного перетворення машинного коду у асемблер. Дизасембльований текст найбільш близький до машинного коду — хіба що числові коди замінені символічними іменами команд та процедур. Тому цей текст дуже корисний для аналізу програми.

He зважаючи вихідні тексти асемблері, на те. ЩО ΜИ пишемо на відрізнятися дизасембльований текст може суттєво від написаного програмістом вихідного тексту. Наприклад, замість invoke буде декілька push, а потім call. Чим більше програміст використовує високорівневих синтаксичних конструкцій певної версії асемблера, тим більше вихідний текст програми відрізнятиметься від дизасембльованого (рис. 15).




Рис. 15. Середовище Visual Studio у режимі налагодження програми

Для того, щоб простежити роботу програми по крокам натискуйте клавішу F10 або клавішу F11. Чим вони відрізняються? При натискуванні F11 (Шаг с заходом) налагоджувач входить у процедури — так можна увійти у процедуру MessageBoxA і простежити її код та процес виконання. При натискуванні F10 (Шаг с обходом) у середину процедур налагоджувач не входить — поточна процедура виконується немов би за один крок.

Варіанти завдання

Усім студентам необхідно запрограмувати:

- початкове діалогове вікно-вітання від автора програми;
- виконання команди CPUID з параметрами 0, 1, 2 а також 80000000h, 80000001h, 80000002h, 80000003h, 80000004h, 80000005h та 80000008h. Кожний результат виконання CPUID команди потрібно виводити у окремому діалоговому вікні. Якщо результати CPUID утворюють текстові дані, то виводити їх як рядки тексту.

Отримати дизасембльований код і проаналізувати його.

Пояснити значення N-го біту кожного результату команди CPUID, де N – номер студента у списку у журналі. Для пояснення використати документ "Intel® 64 and IA-32 Architectures Software Developer's Manual Volume 2A: Instruction Set Reference", доступний на сайті фірми Intel.

Зміст звіту:

- 1. Титульний лист
- 2. Завдання
- 3. Роздруківка тексту програми, основних результатів. Примітка. Задля дотримання всілякої конфіденційності у звіті не треба наводити роздруківку усіх кодів результатів, отриманих командами CPUID. Достатньо продемонструвати виконання програми на комп'ютері.
- 4. Аналіз, коментар вихідного тексту та дизасембльованого коду
- 5. Висновки

Контрольні питання:

- 1. Як розпочати проект на асемблері у середовищі MS Visual Studio?
- 2. Як додати файл вихідного тексту на асемблері у проект MS Visual Studio?
- 3. Як отримати виконуємий файл програми у середовищі MS Visual Studio?
- 4. Як налаштовувати конфігурації Debug та Release?
- 5. Як простежити роботу програми у налагоджувачі MS Visual Studio?
- 6. Що таке дизасембльований код?
- 7. Як можна проаналізувати машинний код?
- 8. Чим відрізняється dword ptr від byte ptr і навіщо вони потрібні?
- 9. Як задати параметр команді CPUID?